
Recall
• Substructural Types
• The structural rules – Exchange, Weakening, Contraction – make

a variable a fact
• Dropping them yields a spectrum: affine, relevant, linear, ordered
• A linear λ-calculus tags types lin P | un P ; splitting the context +

no weakening ⇒ used exactly once
• Rust ownership is affine: one owner; moving invalidates the old name

Today: effect systems – typing not just what a value is, but what a
computation does.

A Type Says What, Not What It Does
An ordinary type gives the shape of the result – never the side effects
along the way. Koka1 puts them in the type:

fun square( x : int ) : total int // `total` = pure, no effect
x * x

fun risky( x : int ) : <exn,console> int // may throw AND print
if x < 0 then throw("neg") else { println(x); x }

The primitive effects we track:

exn (throw) io (I/O) st (state) div (non-termination)

So, the judgment becomes type-and-effect:

Γ ⊢ e : τ ! ε

which means that “e has type τ with effect ε”.

1https://koka-lang.github.io/

An Effect Is a Set of Things That May Happen
An effect ε is a set of primitive labels; the empty set ∅ means pure.
Effects combine by union and are ordered by inclusion:

∅ ⊆ {exn} ⊆ {exn, io} ⊆ {exn, io, st} ⊆ · · ·

So (P({exn, io, st, div}), ⊆, ∪) is a join semilattice – exactly the
structure we need to accumulate effects.

Effects Live on the Arrow
A function definition does nothing; the effect happens when it is called.

So, the effect is latent on the arrow:

τ1
ε−→ τ2 (Koka writes this int -> <exn> int)

Types τ ::= bool | int | τ1
ε−→ τ2

Effects ε ::= a set ⊆ {exn, io, st, div}
Terms e ::= x | λx:τ. e | e e | throw e | try e catch e

AAA551 @ Korea University Lecture 26 – Effect Systems June 9, 2026 8 / 24



Effects Live on the Arrow
A function definition does nothing; the effect happens when it is called.

So, the effect is latent on the arrow:

τ1
ε−→ τ2 (Koka writes this int -> <exn> int)

Types τ ::= bool | int | τ1
ε−→ τ2

Effects ε ::= a set ⊆ {exn, io, st, div}
Terms e ::= x | λx:τ. e | e e | throw e | try e catch e

AAA551 @ Korea University Lecture 26 – Effect Systems June 9, 2026 8 / 24



Typing Rules: Values Are Pure, Application Unions
Values produce no effect; T-Abs discharges the body’s effect ε onto
the arrow:

x:τ ∈ Γ
Γ ⊢ x : τ ! ∅

T-Var
Γ, x:τ1 ⊢ e : τ2 ! ε

Γ ⊢ λx:τ1. e : (τ1
ε

−−→ τ2) ! ∅
T-Abs

Application unions three effects – 1) the function, 2) the argument, and
3) the latent effect ε released at the call site:

Γ ⊢ e1 : (τ1
ε

−−→ τ2) ! ε1 Γ ⊢ e2 : τ1 ! ε2

Γ ⊢ e1 e2 : τ2 ! ε1 ∪ ε2 ∪ ε
T-App

Typing Rules: Raising and Handling
throw adds the exn effect (and may have any result type – it never
returns normally):

Γ ⊢ e : τ ! ε

Γ ⊢ throw e : τ ′ ! ε ∪ {exn}
T-Throw

A handler is the only way to remove an effect: try subtracts exn from
the protected expression.

Γ ⊢ e1 : τ ! ε1 Γ ⊢ e2 : τ ! ε2

Γ ⊢ try e1 catch e2 : τ ! ( ε1 \ {exn}) ∪ ε2
T-Try

Primitives introduce effects, handlers eliminate them. Subtracting exn
is well-defined whether or not ε already contains it.

Subeffecting: Over-Approximating Is Always Safe
Claiming more effects than really occur is sound:

Γ ⊢ e : τ ! ε ε ⊆ ε′

Γ ⊢ e : τ ! ε′ T-Sub

So, a conditional just takes the join of its branches.
Γ ⊢ e1 : bool ! ε1 Γ ⊢ e2 : τ ! ε2 Γ ⊢ e3 : τ ! ε3

Γ ⊢ if e1 e2 e3 : τ ! ε1 ∪ ε2 ∪ ε3
T-If

A Worked Example

fun safe-div( a : int, b : int ) : exn int
if b == 0 then throw("div by zero") else a / b

fun run() : int
with catch( fn(err) { 0 } ) // handler discharges `exn`: return 0
safe-div(10, 0)

throw(. . .) : int ! {exn} (T-Throw)
if (b=0) . . . (a/b) : int ! {exn} (T-If)

safe-div : (int, int) {exn}−−−→ int (T-Abs)
safe-div(10, 0) : int ! {exn} (T-App)
catch(. . .) : int ! ∅ (T-Try subtracts exn)

Note that Koka’s built-in `/` is total (x/0 = 0, it never throws).

Algebraic Effects & Handlers: the Vocabulary
In Koka, we can define user-defined algebraic effects and handlers:

// `effect` : declare a new effect whose name is `ask`
effect ask

// `ctl` : declare a control operation (capture continuation) `ask`
ctl ask() : int

// or simply `effect ctl ask() : int` for a single operation

fun run() : int
// `with` (`handler`) : install a handler in the rest of the scope
with handler {

// `return` : transform the final result value (here: unchanged)
return(x) x
// `resume` : continue the suspended computation with a value
ctl ask() { resume(21) }

}
// or simply `with ctl ask() { resume(21) }` when there is no `return`

// ask() -- suspends, runs handler, resumes with 21, so 21 * 2 = 42
ask() * 2

How Often You Resume Decides the Effect
The resume restarts the rest of the computation after the operation.

How many times the handler calls resume represents different effects:

# resume calls the rest of the computation . . . effect
0 is dropped – abort, jump out exception
1 runs exactly once – the ordinary case reader, state, output

≥ 2 re-runs – all branches, or stream items nondeterminism, generator

Resume 0 Times – Exceptions
If the handler never calls resume, the suspended computation is
discarded – control jumps straight out:

effect ctl raise( msg : string ) : a

fun safe-div( a, b ) : raise int
if b == 0 then raise("div by zero") else a / b

fun try-div( a, b ) : int
with ctl raise(msg) { 0 } // NO `resume` -> drop the rest, return 0
safe-div(a, b) // try-div(10, 0) => 0

Resume 1 Time – State
Resuming exactly once with a chosen value is the ordinary case.

Two operations over a mutable cell give state:

effect state
ctl get() : int
ctl put( x : int ) : ()

fun with_state( init, action )
var s := init
with handler

ctl get() resume(s) // resume with the current s
ctl put(x) { s := x; resume(()) } // update s, then resume with ()

action()

fun tick() : state int
val n = get(); put(n + 1); n

fun run() : int
with_state(10) { tick() + tick() } // 10 + 11 = 21

Resume ≥ 2 Times – Nondeterminism
A handler may call resume more than once for a single operation –
forking the computation to explore every outcome:

effect ctl flip() : bool

fun coins() : flip int
(if flip() then 1 else 0) + (if flip() then 1 else 0)

fun amb(action) : list<int>
with handler

return(x : int) [x] // each finished run produces a singleton list
ctl flip() resume(True) ++ resume(False) // resume BOTH ways

action()

fun run()
amb(coins) // [2,1,1,0] (all four outcomes)

where return converts the final result of the handled computation into
another type – here, an integer x into a singleton list [x].

Resume ≥ 2 Times – Generators
A generator resumes once per item, so over the whole run the
continuation advances many times – streaming values out lazily:

effect ctl yield( x : int ) : ()

fun upto( i : int, n : int ) : <div,yield> () // recursion adds `div`
if i < n then { yield(i); upto(i + 1, n) }

fun to-list( action ) : <div> list<int>
var xs := []
with ctl yield(x){

xs := Cons(x, xs); resume(()) // record x, then resume
}
action()
xs.reverse

fun run() : <div> list<int>
to-list({ upto(3,7) }) // [3,4,5,6]

Algebraic Effects, Formally
The core idea (Plotkin & Pretnar, 2009): split terms into values v and
computations c, and add just two new forms:

Values v ::= x | () | true | false | λx. c | · · ·
Comp. c ::= return v | op(v; y.c) | v1 v2 | with h handle c | · · ·
Handlers h ::= { return x 7→ cr, [ op(x; k ) 7→ cop ]op }

• Perform an operation, op(v; y.c) – call op with argument v, bind its
result to y, and continue as c. So y.c is the rest of the
computation.

• Handle it, with h handle c – the handler h gives a return clause
(for a final value) and, per operation, a clause cop using the argument
x and the resumption k (this is resume).

Handling: the Resumption k Is resume
Handling has just two rules – a return runs the return clause; an
operation runs its clause, with k bound to “the rest, re-handled”:

with h handle (return v) → cr[ x 7→ v ]

with h handle (op(v; y.c)) → cop

[
x 7→ v, k 7→ λy. with h handle c

]

That second rule is the whole story: op is intercepted, and k – the
captured continuation – is handed to the clause. How often cop calls k
decides the effect:

0 (drop) ⇒ exception 1 ⇒ state, reader ≥ 2 ⇒ nondet., generator

Effect Systems in Practice
Tracking what a computation does – not just its result – now appears in
many forms:

System Mechanism Tracks
Java throws clauses checked exceptions
Haskell monads (IO, State) effects in the type ctor
Koka row of effect labels exn, div, st, . . .
Eff / OCaml 5 algebraic eff. + handlers user-defined operations
Scala ZIO ZIO[R,E,A] errors E, capabilities R
Rust async, unsafe, const coarse-grained effects

The same recipe everywhere: enrich the judgment to Γ ⊢ e : τ ! ε, union
effects as they accumulate, and let handlers discharge them.

Next Lecture
• Course Review

